

ORONER: An Ontology for Representing inter-Organizational NETworks Resilience.

A building block towards adaptive network reconfiguration

Mohamed Ayoub Laouni^{1,2}[0009-0007-4603-0060],
Marisnel Olivares¹[0000-0002-0923-1660], Éric Saux¹[0000-0002-6685-6106], and
José-Luis Vilchis-Medina³[0000-0001-5724-5920]

¹ Naval Academy Research Institute (IRENav), Brest Cedex 09, France

²LEGO, University of Western Brittany, 20 avenue Victor Le Gorgeu, CS 93837,
29238 Brest Cedex 3, France

³ Lab-STICC, Robex, ENSTA-Bretagne, Brest, 29200, France
{mohamed.laouni,marisnel.olivares,eric.saux}@ecole-navale.fr,
jose.vilchis@ensta-bretagne.fr

1 Competencies questions

Namespaces which are used in the queries are:

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX event: <http://purl.org/NET/c4dm/event.owl#>
PREFIX geo: <http://www.opengis.net/ont/geosparql#>
PREFIX time: <http://www.w3.org/2006/time#>
PREFIX oroner: <http://www.semanticweb.org/2022/9/OrganisationResilienceOntology#>
```

1.1 General competencies questions about organizations

CQ1: How can we characterize an organization (e.g. hospital_Quimper)?

```
SELECT ?o1 ?o2 ?m1 ?r1 ?h1 #?h2 ?n1
WHERE {
    ?o1 rdf:type oroner:Organization;
    oroner:hasInteraction ?o2;
    oroner:hasMission ?m1;
    oroner:hasResource ?r1;
    oroner:hasHead ?h1.
    # oroner:isMemberofInterOrganizationNetwork ?n1.
    # ?h2 oroner:hasChief ?h1.
    FILTER (regex(str(?o1), "hospital_Quimper"))
}
ORDER BY ?o1
```

CQ2: Who makes the decisions in an organization?

```
SELECT ?o1 ?h1 #?h2
WHERE {
    ?o1 rdf:type oroner:Organization; oroner:hasHead ?h1.
    #?h2 oroner:hasChief ?h1.
}
```

1.2 Competencies questions about resources

CQ3: What are the resources and their quantity held by the organizations?

```
SELECT ?x ?r ?t ?v ?q
WHERE {
    ?x rdf:type oroner:Organization; oroner:hasResource ?r.
    ?r rdf:type ?t.
    {?r oroner:hasValue ?v.}
    UNION
    {?r oroner:hasQuantity ?q}
}
```

CQ4: What are the capabilities possessed by a human agent in an organization and its relationships with other human agents?

```
SELECT ?p1 ?hc1 ?p2
WHERE {
    ?p1 rdf:type oroner:HumanResource.
    {?p1 oroner:hasHumanCapability ?hc1.}
    UNION
    {?p1 oroner:hasRelationWith ?p2.}
}
```

CQ5: Which chiefs are boundary spanners or have a "Participative" leadership style or "Proactive" human skills?

```
SELECT ?p1 ?ls1 ?hc1
WHERE {
    ?p1 rdf:type oroner:Chief.
    {
        ?p1 oroner:hasLeadershipStyle ?ls1; oroner:hasHumanCapability ?hc1.
        FILTER (regex(str(?ls1), "Participative") || regex(str(?hc1), "Proactive"))
    }
    UNION
    {
        ?p1 rdf:type oroner:BoundarySpanner.
    }
}
```

1.3 Competencies questions about missions

CQ6: What are the missions of organizations by type?

```
SELECT ?x ?m ?t WHERE {
    ?x rdf:type oroner:Organization.
    ?x oroner:hasMission ?m.
    ?m rdf:type ?t.
}
GROUP BY ?x ?m ?t
ORDER BY ?x ?m ?t
```

CQ6.1: Which organizations are treating Covid 19 patients?

```
SELECT ?x ?m ?t
WHERE {
    ?x rdf:type oroner:Organization.
    ?x oroner:hasMission ?m.
    ?m rdf:type ?t.
    FILTER (regex(str(?t), "CaringCovid"))
}
```

1.4 Competencies questions about crisis

CQ7: How can we characterize (factors, impacts, temporal interval, spatial extension, etc.) a crisis?

```
SELECT ?c1 ?f1 ?d1 ?o1 ?p1 ?geo1 ?hgeo1 ?t1 ?t1start ?ht1start ?t1end ?ht1end
WHERE {
    ?c1 rdf:type oroner:Crisis; event:factor ?f1; event:product ?d1;
    oroner:hasImpactOn ?o1; event:place ?p1; event:time ?t1.
    ?p1 geo:hasGeometry ?geo1.
    ?geo1 geo:asWKT ?hgeo1.
    ?t1 time:hasBeginning ?t1start; time:hasEnd ?t1end.
    ?t1start time:inXSDDate ?ht1start.
    ?t1end time:inXSDDate ?ht1end.
}
```

CQ8: What are the damages caused by a crisis?

```
SELECT ?c1 ?d1 ?t1
WHERE {
    ?c1 rdf:type oroner:Crisis; event:product ?d1.
    ?d1 rdf:type ?t1.
}
```

CQ9: What is the crisis management status of an organization before, during and after the crisis?

```
SELECT DISTINCT ?o1 ?c1o1 ?c1 ?val_bcs1 ?val_dcs1 ?val_acs1
WHERE {
    ?o1 rdf:type oroner:Organization; oroner:isImpactedBy ?c1o1.
    ?c1o1 oroner:belongsToCrisis ?c1; oroner:hasCrisisManagementStatus ?cms1.
    {?c1o1 oroner:hasBeforeCrisisManagementStatus ?val_bcs1.}
    UNION
    {?c1o1 oroner:hasDuringCrisisManagementStatus ?val_dcs1.}
    UNION
    {?c1o1 oroner:hasAfterCrisisManagementStatus ?val_acs1.}
}
```

1.5 Competencies questions about interactions

CQ10: What are the possible types of interactions between organizations?

```
SELECT ?p1 ?p2 WHERE {
    ?p1 rdf:type owl:ObjectProperty.
    ?p2 rdf:type owl:ObjectProperty.
    ?p1 rdfs:subPropertyOf ?p2.
    FILTER (str(?p2) = "hasInteraction")
}
```

CQ11: Which organizations are in interaction in an inter-organizational network?

```
SELECT ?n ?o1 ?o2 ?i1 ?t1
WHERE
{
    ?n rdf:type oroner:InterOrganizationNetwork;
    oroner:hasOrganizationMember ?o1; oroner:hasOrganizationMember ?o2.
    ?i1 oroner:hasParticipant ?o1; oroner:hasParticipant ?o2 ;
    oroner:hasType ?t1.
    FILTER (str(?o1) != str(?o2))
}
```

CQ12: Which organizations interacted in the past?

```

SELECT ?o1 ?interaction1 ?type1 ?interval ?intervalstart ?xsdTime1
?intervalend ?xsdTime2
WHERE
{
    ?interaction1 oroner:hasParticipant ?o1.
    ?interaction1 time:hasTime ?interval; oroner:hasType ?type1.
    ?interval time:hasBeginning ?intervalstart ; time:hasEnd ?intervalend.
    ?intervalstart time:inXSDDate ?xsdTime1.
    ?intervalend time:inXSDDate ?xsdTime2.
}

```

CQ13: Which chiefs are members of the same meta-organization?

```

SELECT ?m1 ?x1 ?o1 ?x2 ?o2
WHERE {
    ?x1 oroner:isHeadOf ?o1.
    ?x2 oroner:isHeadOf ?o2.
    ?x1 oroner:isMemberOfMetaOrganization ?m1.
    ?x2 oroner:isMemberOfMetaOrganization ?m1.
    FILTER (str(?o1)!=str(?o2))
}

```

CQ14: Which organizations cooperate in the network?

```

SELECT ?o1 ?p ?o2
WHERE {
    ?o1 rdf:type oroner:Organization.
    ?o2 rdf:type oroner:Organization.
    ?o1 oroner:cooperate ?o2 .
}

```

CQ15: Which organizations are in conflict with and who has authority over them?

```

SELECT ?o1 ?o2 ?o3
WHERE {
    ?o1 rdf:type oroner:Organization.
    ?o2 rdf:type oroner:Organization.
    ?o3 rdf:type oroner:Organization.
    ?o1 oroner:isInConflict ?o2 .
    ?o3 oroner:hasVerticalInteraction ?o2.
}

```

CQ16: Which organizations are in conflict and which are under their control?

```

SELECT ?o1 ?o2 ?o3
WHERE {
    ?o1 rdf:type oroner:Organization.
    ?o2 rdf:type oroner:Organization.
    ?o3 rdf:type oroner:Organization.
    ?o1 oroner:isInConflict ?o2 .
    ?o2 oroner:hasVerticalInteraction ?o3.
}

```

CQ17: Which organizations have horizontal interactions and have care missions?

```

SELECT ?o1 ?o2 ?m1 ?m2
WHERE {
    ?o1 rdf:type oroner:Organization.
    ?o2 rdf:type oroner:Organization.
    ?o1 oroner:hasHorizontalInteraction ?o2.
    ?o1 oroner:hasMission ?m1.
    ?o2 oroner:hasMission ?m2.
    FILTER(CONTAINS(xsd:string(?m1),"caring")
    && CONTAINS(xsd:string(?m2),"caring"))
}

```

CQ18: Which organizations have horizontal interactions and share the same missions?

```

SELECT ?o1 ?o2 ?m1 ?m2
WHERE {
    ?o1 rdf:type oroner:Organization.
    ?o2 rdf:type oroner:Organization.
    ?o1 oroner:hasHorizontalInteraction ?o2.
    ?o1 oroner:hasMission ?m1.
    ?o2 oroner:hasMission ?m2.
    FILTER(SUBSTR(xsd:string(?m1), 64,8) = SUBSTR(xsd:string(?m2), 64,8))
}

```

2 Rules

The goal of the implemented rules is to determine if there are possible resource exchanges between two organizations.

R1: If ?x is a boundary spanner between two different organizations, then there are possible resource exchanges between them.

```

orga:BoundarySpanner(?x) ^ orga:isHumanResourceOf(?x, ?y)
^ orga:isHumanResourceOf(?x, ?z) ^ differentFrom(?y, ?z)
-> orga:PossibleResourceExchange(?y, ?z)

```

R2: If ?x1 is the head of an organization ?y1 and a member of a meta-organization ?z and if ?x2 is the head of an organization ?y2 and a member of the same meta-organization ?z, then there are possible resource exchanges between the two organizations ?y1 and ?y2.

```

orga:isMemberOfMetaOrganization(?x1, ?z)
^ orga:isMemberOfMetaOrganization(?x2, ?z) ^ orga:isHeadOf(?x1, ?y1)
^ orga:isHeadOf(?x2, ?y2)
-> orga:PossibleResourceExchange(?y1, ?y2)

```

R3: If ?x1 is a top level management of an organization ?y1 and a member of a meta-organization ?z and if ?x2 is a top level management of an organization ?y2 and a member of the same meta-organization ?z, then there are possible resource exchanges between the two organizations ?y1 and ?y2.

```
orga:isMemberOfMetaOrganization(?x1, ?z)
^ orga:isMemberOfMetaOrganization(?x2, ?z) ^ orga:TopLevelManagement(?x1)
^ orga:isHumanResourceOf(?x1, ?y1) ^ orga:TopLevelManagement(?x2)
^ orga:isHumanResourceOf(?x2, ?y2)
-> orga:PossibleResourceExchange(?y1, ?y2)
```

R4: If ?x is the head of an organization ?y and a member of a meta-organization ?z to which an organization ?w belongs, then there are possible resource exchanges between the two organizations ?y and ?w.

```
orga:isMemberOfMetaOrganization(?x, ?z) ^ orga:isComposedOfOrganization(?z, ?w)
^ orga:isHeadOf(?x, ?y)
-> orga:PossibleResourceExchange(?y, ?w)
```

R5: If ?x is a top level management of an organization ?y and a member of a meta-organization ?z to which an organization ?w belongs, then there are possible resource exchanges between the two organizations ?y and ?w.

```
orga:isMemberOfMetaOrganization(?x, ?z) ^ orga:TopLevelManagement(?x)
^ orga:isHumanResourceOf(?x, ?y) ^ orga:isComposedOfOrganization(?z, ?w)
-> orga:PossibleResourceExchange(?y, ?w)
```